
13 November, 2012 - 16:01 — SAS-Fidi

Ich habe ein Macro, das alle Dateien eines Verzeichnisses nach Excel schreibt. Nur leider nimmt es
die Variablenformate nicht mit. Statt "no" steht da also z.B. 0 und statt "yes" eine 1.

Kann ich das Makro so ändern, dass die Formate mit übertragen werden? Ich dachte daran, vor der
Erzeugnung der Excel-Dateien die Ausprägungen der Variablen mit ihren Formaten zu tauschen,
aber ich bekomme es nicht hin.

Hier das Macro:

%MACRO EXPORT_EXCEL();

PROC DATASETS LIB=WORK MEMTYPE=DATA KILL;
RUN;
DM "CLEAR LOG;";

 * Reads dataset names;
 PROC CONTENTS DATA=&libout.._ALL_ OUT=DSET_CONT1 NOPRINT Details; *read all informations about datasets of the library;
 RUN;

 DATA DSET_CONT1;
 SET DSET_CONT1;
  IF INDEX(LABEL,'.') GT 0 THEN RE = 1 ; ELSE RE = 0; 
  LABEL = COMPRESS(LABEL,'.');
  IF NAME EQ "COVAL1" THEN DO;
   LABEL = "Comment2";
   RE = 1;
  END;
  IF NAME EQ "TSVAL1" THEN DO;
   LABEL = "Parameter Value2";
   RE = 1;
  END;
 RUN;

 PROC SORT DATA=DSET_CONT1;
  BY MEMNAME NAME;
 RUN;

 DATA DSET_CONT;
  SET DSET_CONT1;
  BY MEMNAME NAME;
  IF FIRST.MEMNAME;
 RUN;

 * Deletes additional Label [...];
 DATA DSET_CONT;
  SET DSET_CONT;
  IF MEMLABEL EQ "" THEN MEMLABEL = MEMNAME;
  ELSE DO;
   X = INDEX(MEMLABEL,'[');
   IF X NE 0 THEN MEMLABEL = SUBSTR(MEMLABEL,1,X-2);
  END;
  
 RUN;

 * Writes all necessary values into macro variables;
 PROC SQL NOPRINT;
  SELECT COUNT(DISTINCT MEMNAME) INTO :dset_anz SEPARATED BY ' ' FROM DSET_CONT; *read how much observations exist;
  SELECT MEMNAME, MEMLABEL INTO :dset_s1-:dset_s&dset_anz, :dset_l1-:dset_l&dset_anz 
   FROM DSET_CONT; *read the short and long label of each dataset;
  SELECT MAX(LAENGE) INTO :dset_max SEPARATED BY ' ' FROM 
   (SELECT LENGTH(MEMNAME) AS LAENGE FROM DSET_CONT); *read the max length for dataset label;
 QUIT;

 * Export of Excel files;

 %DO i=1 %TO &dset_anz;

   DATA DSET_CONT2;
    SET DSET_CONT1;
    IF MEMNAME EQ "&&dset_s&i";
    
   RUN;
   DATA &&dset_s&i;
    SET &libout..&&dset_s&i;
   
   RUN;

   PROC SQL NOPRINT;
    SELECT COUNT(DISTINCT NAME) INTO :v_anz SEPARATED BY ' ' FROM DSET_CONT2;
    SELECT NAME, LABEL,RE INTO :v_s1-:v_s&v_anz, :v_l1-:v_l&v_anz ,:r_l1-:r_l&v_anz 
     FROM DSET_CONT2; *read the short and long label of each dataset;  
   QUIT;

   %DO j=1 %TO &v_anz;

   
   %PUT &&r_l&j;
   %PUT &&v_s&j;
   %PUT &&v_l&j;

    DATA &&dset_s&i;
    SET &&dset_s&i;
    %IF &&r_l&j EQ 1 %THEN %DO; LABEL &&v_s&j = "&&v_l&j"; %END;
    RUN;

   %END;
 
  
   PROC EXPORT DATA= &&dset_s&i
    OUTFILE= "&OUTPATH\&&dset_s&i...xls" 
    DBMS=EXCELCS LABEL REPLACE;
    SHEET="&&dset_s&i"; 
   RUN;

 %END;

%MEND EXPORT_EXCEL;

VG SAS-Fidi

14 November, 2012 - 09:31 — HansKneilmann

Kommentare

sorry
Hallo Elfriede,
sorry, bei mir kommt nach PROC EXPORT DATA=... nur

ERROR: DBMS type EXCELCS not valid for export..

Es wäre sehr hilfreich, wenn bei so einem Problem (ohne Rätsel raten) klar wäre, welche Maco-
Variablen als Input an den Macro gehen.
Das kann man gut so lösen:

%MACRO EXPORT_EXCEL(libout=, OUTPATH=);
/* hier kommt der unveränderte Code */
%MEND EXPORT_EXCEL;

Dazu passend macht ein Aufruf-Beispiel das (Helfer-) Leben einfacher:

%EXPORT_EXCEL(OUTPATH=c:\tmp, libout=dmworkOI);

Und zu guter Letzt: Den Code mit <sascode> und </sascode> einschließen, dann ist er besser
lesbar, sogar ohne den Browser XL-breit zu ziehen (was ich hasse und nie mache).



14 November, 2012 - 17:33 — JanHeuer

Gruß
Hans Kneilmann, Schäfer Shop GmbH (SSI)

mit sql dictionary code generieren und ausführen
Hallo SAS-Fidi,
ich würde das SQL-Dictionary dafür nehmen.
Mit den Informationen aus dictionary.columns kann man einen Sql-Befehl erstellen und diesen dann
ausführen.
Anschließend exportiert man die Tabelle, wobei alle formatierten Werte als Text exportiert werden.

%macro ExportExcelMitFormat(lib=WORK,mem=,outfile=,sheet=);
  %local sqlstring;
  Proc sql noprint;
    select case 
        when Format is not null then "put(" || name ||"," || Format ||") as " || name 
        else name 
    end 
    into :sqlstring separated by ","
    from dictionary.columns
    where upcase("&lib.")=libname and upcase("&mem.")=memname
    ;
    %let sqlstring  = Create table Zwischentabelle as select &sqlstring. from &lib..&mem.;
    &sqlString.;
  quit;
  PROC EXPORT 
   DATA     = Zwischentabelle 
   OUTFILE  = &Outfile.
   DBMS     =  Excel
   REPLACE
  ;
   Sheet    = &sheet.;
  RUN;
%mend;

Als Beispiel exportiere ich eine formatierte Sashelp.class.
Im ersten Beispiel werden die Variablen Weight und Height nicht formatiert und als Zahl exportiert.
Im zweiten Beispiel werden die beiden Variablen formatiert als Text exportiert.

Proc format; 
  value Age
    low - 12 = "kein Teenager"
    13 - high = "Teenager"
  ;
  value $sex
    "F" = "weiblich"
    "M" = "männlich"
  ;
run;
Data class;
  set sashelp.class;
  format sex $sex. age age. ;
run;
%ExportExcelMitFormat(
  mem     = class
, outfile = "c:\temp\test.xls"
, sheet   = "class"
);
Data class;
  set sashelp.class;
  format sex $sex. age age. weight height commax12.2;
run;
%ExportExcelMitFormat(
  lib         = work
, mem     = class
, outfile = "c:\temp\test.xls"
, sheet   = "class1"
);

Viele Grüße

Jan

